Simple Pendulum DAE

```
> with(plots):
> DAE1 := m*diff(x(t),t,t)+2*x(t)*lambda(t) ;
  DAE2 := m*diff(y(t),t,t)+2*y(t)*lambda(t)+m*g ;
  DAE3 := x(t)^2+y(t)^2-r^2 ;
```

$$DAE1 := m \left(\frac{d^2}{dt^2} x(t) \right) + 2\, x(t)\, \lambda(t)$$

$$DAE2 := m \left(\frac{d^2}{dt^2} y(t) \right) + 2\, y(t)\, \lambda(t) + m\, g$$

$$DAE3 := x(t)^2 + y(t)^2 - r^2 \quad (1)$$

Convert to first order DAE

```
> DAE1 := diff(x(t),t)-u(t) ;
  DAE2 := diff(y(t),t)-v(t) ;
  DAE3 := m*diff(u(t),t)+2*x(t)*lambda(t) ;
  DAE4 := m*diff(v(t),t)+2*y(t)*lambda(t)+m*g ;
  ALG1 := x(t)^2+y(t)^2-r^2 ;
```

$$DAE1 := \frac{d}{dt} x(t) - u(t)$$

$$DAE2 := \frac{d}{dt} y(t) - v(t)$$

$$DAE3 := m \left(\frac{d}{dt} u(t) \right) + 2\, x(t)\, \lambda(t)$$

$$DAE4 := m \left(\frac{d}{dt} v(t) \right) + 2\, y(t)\, \lambda(t) + m\, g$$

$$ALG1 := x(t)^2 + y(t)^2 - r^2 \quad (2)$$

```
> SOLS := solve( {DAE||(1..4)},diff({x(t),y(t),u(t),v(t)},t)) ;
```

$$SOLS := \left\{ \frac{d}{dt} u(t) = -\frac{2\, x(t)\, \lambda(t)}{m},\ \frac{d}{dt} v(t) = -\frac{2\, y(t)\, \lambda(t) + m\, g}{m},\ \frac{d}{dt} x(t) = u(t),\ \frac{d}{dt} y(t) = v(t) \right\} \quad (3)$$

Evaluate, index, reduce to ODE

```
> D1 := diff(ALG1,t) ;
```

$$D1 := 2\, x(t) \left(\frac{d}{dt} x(t) \right) + 2\, y(t) \left(\frac{d}{dt} y(t) \right) \quad (4)$$

```
> ALG2 := subs(SOLS,D1)/2 ;
```

$$ALG2 := x(t)\, u(t) + y(t)\, v(t) \quad (5)$$

```
> D2 := diff(ALG2,t) ;
```

$$D2 := \left(\frac{d}{dt} x(t) \right) u(t) + x(t) \left(\frac{d}{dt} u(t) \right) + \left(\frac{d}{dt} y(t) \right) v(t) + y(t) \left(\frac{d}{dt} v(t) \right) \quad (6)$$

```
> ALG3 := algsubs(x(t)^2+y(t)^2=r^2,collect(subs(SOLS,D2),[lambda,m])
```

```
) ;
```

$$ALG3 := -\frac{2\,\lambda(t)\,r^2}{m} - y(t)\,g + u(t)^2 + v(t)^2 \quad \textbf{(7)}$$

```
> solve( ALG3,{lambda(t)} ) ;
```

$$\left\{\lambda(t) = -\frac{1}{2}\,\frac{\left(-u(t)^2 - v(t)^2 + y(t)\,g\right)m}{r^2}\right\} \quad \textbf{(8)}$$

```
> D3 := diff(ALG3,t) ;
```

$$D3 := -\frac{2\left(\frac{d}{dt}\,\lambda(t)\right)r^2}{m} - \left(\frac{d}{dt}\,y(t)\right)g + 2\,u(t)\left(\frac{d}{dt}\,u(t)\right) + 2\,v(t)\left(\frac{d}{dt}\,v(t)\right) \quad \textbf{(9)}$$

```
> ALG4 := op(collect(expand(solve( subs(SOLS,D3), {diff(lambda(t),t)}
  )),[lambda,r])) ;
```

$$ALG4 := \frac{d}{dt}\,\lambda(t) = \frac{\left(-2\,x(t)\,u(t) - 2\,y(t)\,v(t)\right)\lambda(t)}{r^2} - \frac{3}{2}\,\frac{v(t)\,g\,m}{r^2} \quad \textbf{(10)}$$

```
> ODE5 := lhs(ALG4)-rhs(ALG4) ;
```

$$ODE5 := \frac{d}{dt}\,\lambda(t) - \frac{\left(-2\,x(t)\,u(t) - 2\,y(t)\,v(t)\right)\lambda(t)}{r^2} + \frac{3}{2}\,\frac{v(t)\,g\,m}{r^2} \quad \textbf{(11)}$$

Set initial condition

```
> INI := x(0) = r*cos(theta0), y(0)=r*sin(theta0) ;
```

$$INI := x(0) = r\cos(\theta 0),\, y(0) = r\sin(\theta 0) \quad \textbf{(12)}$$

```
> subs(INI,subs(t=0,ALG1)) ; simplify(%) ;
```

$$r^2\cos(\theta 0)^2 + r^2\sin(\theta 0)^2 - r^2$$

$$0 \quad \textbf{(13)}$$

```
> INI := INI, u(0) = V*sin(theta0), v(0)=-V*cos(theta0) ;
```

$$INI := x(0) = r\cos(\theta 0),\, y(0) = r\sin(\theta 0),\, u(0) = V\sin(\theta 0),\, v(0) = -V\cos(\theta 0) \quad \textbf{(14)}$$

```
> subs(INI,subs(t=0,ALG2)) ;
```

$$0 \quad \textbf{(15)}$$

```
> subs( INI, subs(t=0,ALG3) ) ;
  simplify(solve(%,{lambda(0)})) ;
  INI := INI, op(%) ;
```

$$-\frac{2\,\lambda(0)\,r^2}{m} - r\sin(\theta 0)\,g + V^2\sin(\theta 0)^2 + V^2\cos(\theta 0)^2$$

$$\left\{\lambda(0) = -\frac{1}{2}\,\frac{m\left(r\sin(\theta 0)\,g - V^2\right)}{r^2}\right\}$$

$$INI := x(0) = r\cos(\theta 0),\, y(0) = r\sin(\theta 0),\, u(0) = V\sin(\theta 0),\, v(0) = -V\cos(\theta 0),\, \lambda(0) = -\frac{1}{2}\,\frac{m\left(r\sin(\theta 0)\,g - V^2\right)}{r^2} \quad \textbf{(16)}$$

Setup the data

```
> g      := 9.8 ;
  r      := 1 ;
```

```
m       := 1 ;
theta0  := -Pi/30 ;
V       := 0.1 ;
```

$$g := 9.8$$
$$r := 1$$
$$m := 1$$
$$\theta 0 := -\frac{1}{30}\pi$$
$$V := 0.1 \quad \mathbf{(17)}$$

```
> INI ;
```

$$x(0) = \cos\left(\frac{1}{30}\pi\right), y(0) = -\sin\left(\frac{1}{30}\pi\right), u(0) = -0.1\sin\left(\frac{1}{30}\pi\right), v(0) = -0.1\cos\left(\frac{1}{30}\pi\right), \lambda(0) = 4.900000000\sin\left(\frac{1}{30}\pi\right) + 0.005000000000 \quad \mathbf{(18)}$$

```
> ODE := DAE||(1..4), ODE5 : <%> ;
```

$$\begin{bmatrix} \frac{d}{dt}x(t) - u(t) \\ \frac{d}{dt}y(t) - v(t) \\ \frac{d}{dt}u(t) + 2x(t)\lambda(t) \\ \frac{d}{dt}v(t) + 2y(t)\lambda(t) + 9.8 \\ \frac{d}{dt}\lambda(t) - (-2x(t)u(t) - 2y(t)v(t))\lambda(t) + 14.70000000\,v(t) \end{bmatrix} \quad \mathbf{(19)}$$

```
> example(dsolve) ;
> SOL := dsolve( {ODE, INI}, numeric, method=classical[foreuler],
  output=Array([seq(k/10,k=1..100)]) ) ;
```

$$SOL := \begin{bmatrix} \begin{bmatrix} t & \lambda(t) & u(t) & v(t) & x(t) & y(t) \end{bmatrix} \\ \begin{bmatrix} \textit{100 x 6 Matrix} \\ \textit{Data Type: anything} \\ \textit{Storage: rectangular} \\ \textit{Order: Fortran\_order} \end{bmatrix} \end{bmatrix} \quad \mathbf{(20)}$$

```
> VALS := SOL[2][1] ;
```

$$VALS := \begin{bmatrix} \textit{100 x 6 Matrix} \\ \textit{Data Type: anything} \\ \textit{Storage: rectangular} \\ \textit{Order: Fortran\_order} \end{bmatrix} \quad \mathbf{(21)}$$

```
> X := VALS[1..-1,5] ;
  Y := VALS[1..-1,6] ;
```

$$X := \begin{bmatrix} \text{1 .. 100 } Vector_{column} \\ \text{Data Type: anything} \\ \text{Storage: rectangular} \\ \text{Order: Fortran\_order} \end{bmatrix}$$

$$Y := \begin{bmatrix} \text{1 .. 100 } Vector_{column} \\ \text{Data Type: anything} \\ \text{Storage: rectangular} \\ \text{Order: Fortran\_order} \end{bmatrix} \quad \textbf{(22)}$$

```
> T := VALS[1..-1,1] ;
```

$$T := \begin{bmatrix} \text{1 .. 100 } Vector_{column} \\ \text{Data Type: anything} \\ \text{Storage: rectangular} \\ \text{Order: Fortran\_order} \end{bmatrix} \quad \textbf{(23)}$$

```
> plot( [seq([T[i],X[i]^2+Y[i]^2],i=1..30)], scaling=CONSTRAINED ) ;
```




```
> plot( [seq([X[i],Y[i]],i=1..30)], scaling=CONSTRAINED ) ;
```


1.5
1
0.5
0
−0.5
−1
−1
1
2